

```
100 REM *****
110 REM ***      subroutine: comfort-inkey      ***
120 REM ***      (c) by JEM-SOFT/Germering      ***
130 REM ***                                          ***
140 REM ***      params:                          ***
150 REM ***      iilin% = zeile                    ***
160 REM ***      iicol% = spalte                  ***
170 REM ***      iilen% = eingabelaenge           ***
180 REM ***      iityp% = feldtyp                 ***
190 REM ***      1: alle asc-zeichen             ***
200 REM ***      2: alle asc-zeichen             ***
210 REM ***      iianf%-iiend%                   ***
220 REM ***      3: alle asc-zeichen             ***
230 REM ***      aus iival$                      ***
240 REM ***      8: alle zahlen ohne             ***
250 REM ***      9: alle zahlen mit              ***
260 REM ***      dezimalpunkt                    ***
270 REM ***      iitxt$ = variable zur           ***
```

```

280 REM *** textuebergabe ***
290 REM *** iival$ = alle benuetzbaren ***
300 REM *** zeichen bei typ=3 ***
310 REM *** iianf$ = erstes ***
320 REM *** iient$ = letztes benuetzbares ***
330 REM *** zeichen bei typ=2 ***
340 REM *** ***
350 REM *** intern benuetzte variablen: ***
360 REM *** iica% = spalte aktuell ***
370 REM *** iipos% = position aktuell ***
380 REM *** iiefg% = flag einfuegemodus ***
390 REM *** iipnt% = flag dezimalpunkt ***
400 REM *** iin$ = inkey-zeichen akt. ***
410 REM *** ***
420 REM *** steuervariablen: ***
430 REM *** es ist jeweils der ASCII-code ***
440 REM *** anzugeben. ***
450 REM *** bei inhalt=0 werden standart- ***
460 REM *** werte gesetzt. ***
470 REM *** iifil% = platzhaltezeichen ***
480 REM *** iidl% = delimiterzeichen l ***
490 REM *** iidr% = delimiterzeichen r ***
500 REM *** iicrs% = flag cursor blink ***
510 REM *****
520 GOSUB 1030:iica%=iicol%:iiefg%=0:iipnt%=0
530 IF iifil%=0 THEN iifil%=32:'platzhalter
540 IF iidl%=0 THEN iidl%=246:'delimiter links
550 IF iidr%=0 THEN iidr%=247:'delimiter rechts
560 PRINT CHR$(22);CHR$(1);
570 IF iicol%>1 THEN LOCATE iicol%-1,iilin%:PRINT CHR$(iidl%);
580 IF iicol%+iilen%<80 THEN LOCATE iicol%+iilen%,iilin%:PRINT CHR$(iidr%);
590 PRINT CHR$(22);CHR$(0);
600 LOCATE iicol%,iilin%:PRINT STRING$(iilen%,CHR$(iifil%));
610 LOCATE iicol%,iilin%:PRINT CHR$(22);CHR$(1);:PRINT iitxt$;
620 PRINT CHR$(22);CHR$(0);
630 IF iica%-iicol%=iilen% THEN PRINT CHR$(7);:iica%=iica%-1
640 LOCATE iica%,iilin%:CALL &BB8A
650 iin$=INKEY$:IF iin$<>" " THEN 680 ELSE IF iicrs%=0 THEN 650
660 IF iicrs%>0 THEN CALL &BB8D:iicrs%=-iicrs%:GOTO 650
670 CALL &BB8A:iicrs%=-iicrs%:GOTO 650
680 IF iicrs%<0 THEN iicrs%=-iicrs% ELSE CALL &BB8D
690 IF iin$<CHR$(32) THEN 800
700 IF iin$>CHR$(126) THEN 870
710 IF iica%-iicol%=iilen% THEN 790
720 ON iityp% GOTO 730,740,750,730,730,730,760,770
730 GOSUB 930:iica%=iica%+1:GOTO 630
740 IF ASC(iin$)<iianf% OR ASC(iin$)>iient% THEN 790 ELSE 730
750 IF INSTR(iival$,iin$)=0 THEN 790 ELSE 730
760 IF iin$<"0" OR iin$>"9" THEN 790 ELSE 730
770 IF iin$="." AND iipnt%=0 THEN iipnt%=1:GOTO 730
780 GOTO 760
790 PRINT CHR$(7);:GOTO 630
800 REM steuerzeichen <32
810 IF ASC(iin$)=13 THEN RETURN
820 IF ASC(iin$)=9 THEN iiefg%=-iiefg%+1:GOTO 630:'***einfuegemodus ein/aus
830 IF ASC(iin$)<>16 THEN 790 ELSE iipos%=iica%-iicol%
840 iitxt$=LEFT$(iitxt$,iipos%)+RIGHT$(iitxt$,iilen%-iipos%-1)+" "
850 PRINT RIGHT$(iitxt$,iilen%-iipos%);
860 GOSUB 1000:GOTO 630
870 REM steuerzeichen >126
880 iipos%=iica%-iicol%
890 IF ASC(iin$)=242 THEN IF iipos%>0 THEN iica%=iica%-1:GOTO 630 ELSE 790
900 IF ASC(iin$)=243 THEN IF iipos%+1<iilen% THEN iica%=iica%+1:GOTO 630
ELSE GOTO 790
910 IF ASC(iin$)<>127 THEN 790 ELSE IF iipos%<1 THEN 790
920 iipos%=iipos%-1:iica%=iica%-1:LOCATE iica%,iilin%:GOTO 840
930 PRINT iin$;:iipos%=iica%-iicol%
940 IF iiefg% THEN 980
950 LOCATE iica%,iilin%:PRINT CHR$(22);CHR$(1);
960 PRINT CHR$(iifil%);:PRINT CHR$(22);CHR$(0);
970 iitxt$=LEFT$(iitxt$,iipos%)+iin$+RIGHT$(iitxt$,iilen%-iipos%-1):GOTO 1020
980 iitxt$=LEFT$(iitxt$,iipos%)+iin$+RIGHT$(iitxt$,iilen%-iipos%):GOSUB 1030
990 PRINT RIGHT$(iitxt$,iilen%-iipos%-1);
1000 LOCATE iica%,iilin%:PRINT CHR$(22);CHR$(1);
1010 PRINT STRING$(iilen%-iipos%,CHR$(iifil%));:PRINT CHR$(22);CHR$(0);
1020 RETURN
1030 IF LEN(iitxt$)=iilen% THEN 1060
1040 IF LEN(iitxt$)>iilen% THEN iitxt$=LEFT$(iitxt$,iilen%):GOTO 1060
1050 IF LEN(iitxt$)<iilen% THEN iitxt$=iitxt$+STRING$(iilen%-LEN(iitxt$)," ")
1060 RETURN

```